

```

/** ----- 03 DEM (ELEVATION, SLOPE AND ASPECT) -----
/** FILE NAME:          03_DEM.txt
/** AUTHOR:            RENEE SCHICKER
/** SCRIPT CREATED:    13 AUGUST 2008
/** LAST UPDATED/MODIFIED: 07 DECEMBER 2009
/**
/** The scripts may be supplied in a more readily useable format if the work is acknowledged
/** CONTACT:      Renee_Schicker@hotmail.com
/**
/** STARTS IN:        ARC
/** SCRIPT USED BY:   00_MASTER_org.txt
/** USES THE SCRIPT:  CheckProgGrid.txt      (CREATED: 02 MARCH 2009)
/**
/** INPUT SHAPEFILES: bndsandLake
/** INPUT GRIDS:      north25      region_grid
/** OUTPUT GRIDS:     ElvtnClass   SlopeClass   AspectClass   DEMBounds
/**                  MyBnds
/**
/** TEMP GRIDS MADE:  dem25int     slope      aspect      AspectN
/**                  Elvtn25      slopeint   aspectint   AspectE
/**                  Elvtn50      slope04    aspect0N    AspectS
/**                  Elvtn100     slope08    aspect1N    AspectW
/**                  Elvtn200     slope14    aspect2N    AspectFlat
/**                  Elvtn300     slope20    AspectNE
/**                  Elvtn400     slope30    AspectSE
/**                  Elvtn550     slope45    AspectSW
/**                  Elvtn2800    slope90    AspectNW
/**
/** OUTPUT COVERS:    DEM_Bnd      ElvtnCov    SlopeCov    AspectCov
/** OUTPUT SHAPEFILES: Dem_Bndry
/**
/** FUNCTIONS USED:   &TYPE      &RUN        SETWINDOW   CON
/**                  &CALL      &RETURN     &ROUTINE    GRIDPOLY
/**                  SETMASK    MAPEXTENT   &IF &THEN   [EXIST]
/**                  KILL       QUIT (Q)    ARCSHAPE    SHAPEGRID
/**                  &ELSE      &DO &END
/**
/** PROCESSES EXPLAINED:
/**      EXTRACT      Extracts elevation, slope and aspect from the digital elevation
/**                  model (DEM) as integer (I) values rather than floating point (FLOAT).
/**      ASPECT        Set categorical classes for Flat, N, NE, E,...NW based on orientation
/**                  (numeric - i.e. degree values) by assigning a unique code for each
/**                  class.
/**      SLOPE         Set categorical classes for a range of slopes and assign a unique code
/**                  to each class to match the upper limit of the slope range in that class
/**      ELEVATION      Same idea as for slope but applied to elevation.
/**      DEM_BND        Create a boundary GRID and COVER from the DEM to use later as a
/**                  clip layer.
/**      DEMBnd_Shp     Creates a shapefile version of the output cover
/**      BndandTaupo    Creates a Grid from a copy of the dem_bndry shapefile with Lake
/**                  Taupo cut out. This will be used to clip the output Susceptibility

```

```

/**                                maps.
/**
/** ..... HISTORY.....
/**
/** 04 OCTOBER 2008    DEM_slope.txt singular script made.
/** 15 DECEMBER 2008   8_DEM_slope.txt created as a script to be specifically run from
/**                    master script.
/** 02 MARCH 2009     Added &RUN CheckProgGrid.txt instead of having to rewrite that
/**                    process each time.
/** 9-12 MARCH 2009   Formatted and listed everything, added descriptions and history.
/** 19 MARCH 2009     Added Aspect to be calculated from the DEM.
/** 13 MAY 2009        Region_grid not working, changed name to RegionGrid now works.
/** 20 MAY 2009        Separate Input data and output data directories, so workspace is set
/**                    to a separate output folder, so reduces the chance of deleting input
/**                    data by accident.
/** 26 MAY 2009        Created class_aspect.txt which worked so modified/expanded
/**                    a bit further to make Reclass_DEM.txt .Just need to sort out
/**                    what classes I want.
/** 27 MAY 2009        Completed the slope and elevation sections
/**                    Have also added routines to kill all temp grids.
/** 10 JUNE 2009        Have fused together the content of 12_DEM_slope.txt and
/**                    Reclass_DEM.txt to make 03_DEM.txt. Have reduced the number of
/**                    routines (no more subroutines) and added a new process DEM_Bnd
/**                    which will make a boundary based on the DEM that will be used to
/**                    clip the other variables by. This should mean all data matches up
/**                    really well.
/** 30 SEPTEMBER 2009   Added separate workspaces for each script, so have to add file path
/**                    to find regiongrid which is now in
/**                    D:\Renee_GIS\Output_data\Organised\02_Setup\.
/** 07 DECEMBER 2009   Made the creation of a shapefile process a separate routine.
/**                    Added a shapefile to grid (shapegrid) conversion to create a Grid of
/**                    the final region boundary which will be used to clip the output from
/**                    the statistical analysis (susceptibility maps). The Shape file was
/**                    created in ArcMap by copying the lake feature from the soils layer
/**                    (nzfsl) to a copy of the dem_bndry.shp (created in a previous run of
/**                    03_DEM.txt). The Lake feature was clipped to remove the polygon
/**                    area, leaving empty space. ShapeGrid was found to be the easiest
/**                    and most successful way to convert to Grid
/** *****
/** *****

```

```

&CALL Extract
&CALL Aspect      /* Assign Aspect classes - N, NE, E, SE, S, SW, W, NW, Flat (numeric values)
&CALL Slope       /* Assign Slope classes
&CALL Elevation   /* Assign Elevation classes
&CALL DEM_Bnd
&CALL DEMBnd_Shp  /* create a shapefile of the DEM derived region Boundary

/* create a Grid from a copy of the dem_bndry shapefile with Lake Taupo cut out.
&CALL BndandTaupo
&RETURN

```

```

/*****
&ROUTINE Extract
&TYPE Preparing to extract Elevation, slope and aspect values from the DEM...

/* Need to use GRID so run the associated script to do this
&RUN d:\renee_gis\Scripts\CheckProgGrid.txt
/*set window so slope will work
&TYPE set window to RegionGrid...
SETWINDOW D:\Renee_GIS\Output_data\Organised\02_Setup\RegionGrid
SETMASK D:\Renee_GIS\Output_data\Organised\02_Setup\RegionGrid
&TYPE setmask complete
/*MAPEXTENT D:\Renee_GIS\Output_data\Organised\02_Setup\RegionGrid
/*&TYPE mapextent set

/**----- Elevation -----
&TYPE Calculating Elevation...
/*Convert the DEM to integers to save space and speed up analysis
&IF [EXIST dem25int -grid] &THEN KILL dem25int ALL
&TYPE Converting DEM to integers
dem25int = int(D:\renee_gis\input_data\north25 + .5)

/**----- Slope -----
/* Calculate slope
&TYPE Calculating Slope...
&IF [EXIST slope -grid] &THEN KILL slope ALL
&TYPE Calculating slope
slope = slope(dem25int)

/*Convert slope to integers
&IF [EXIST slopeint -grid] &THEN KILL slopeint ALL
&TYPE Converting slope to integers
slopeint = int(slope)
KILL slope ALL

/**----- Aspect -----
/* Calculate aspect
&TYPE Calculating aspect...
&IF [EXIST Aspect -grid] &THEN KILL Aspect ALL
&TYPE Calculating slope
Aspect = Aspect(dem25int)

/*Convert aspect to integers
&IF [EXIST Aspectint -grid] &THEN KILL Aspectint ALL
&TYPE Converting Aspect to integers
Aspectint = int(Aspect)
KILL Aspect ALL

```

SETMASK off

Q

&RETURN

/*****

&ROUTINE Aspect

&IF [EXIST aspectClass -GRID] &THEN KILL aspectClass all

&R d:\renee_gis\scripts\checkproggrid.txt

&TYPE set window to dem25int...

SETWINDOW dem25int

SETMASK dem25int

&TYPE setmask complete

aspect0N = con(aspectint eq 0, 1, 0) /* North (= 0 degrees)
aspect1N = con(aspectint ge 1 and aspectint lt 23, 1, 0) /* North (>= 1 to < 23 degrees)
aspect2N = con(aspectint ge 338 and aspectint lt 360, 1, 0) /* North (>= 338 to < 360 degrees)

AspectN = con((aspect0N + aspect1N + aspect2N) eq 1, 1, 0) /* N assigned value 1

&TYPE North done

AspectNE = con(aspectint ge 23 and aspectint lt 68, 2, 0) /* NE assigned value 2

&TYPE NE done

AspectE = con(aspectint ge 68 and aspectint lt 113, 3, 0) /* E assigned value 3

&TYPE East done

AspectSE = con(aspectint ge 113 and aspectint lt 158, 4, 0) /* SE assigned value 4

&TYPE SE done

AspectS = con(aspectint ge 158 and aspectint lt 203, 5, 0) /* S assigned value 5

&TYPE South done

AspectSW = con(aspectint ge 203 and aspectint lt 248, 6, 0) /* SW assigned value 6

&TYPE SW done

AspectW = con(aspectint ge 248 and aspectint lt 293, 7, 0) /* W assigned value 7

&TYPE West done

AspectNW = con(aspectint ge 293 and aspectint lt 338, 8, 0) /* NW assigned value 8

&TYPE NW done

AspectFlat = con(aspectint < 0, 9, 0) /* Flat assigned value 9

&TYPE Flat done

AspectClass = (aspectNE + aspectE + aspectSE + aspectS + aspectSW + aspectW + aspectNW +
aspectFlat + aspectN)

&TYPE Aspect Done

SETMASK off

Q

&TYPE Kill Temporary GRIDs

&IF [EXIST aspect0N -GRID] &THEN KILL aspect0N all

&IF [EXIST aspect1N -GRID] &THEN KILL aspect1N all

&IF [EXIST aspect2N -GRID] &THEN KILL aspect2N all

&IF [EXIST aspectNE -GRID] &THEN KILL aspectNE all

```
&IF [EXIST aspectSE -GRID] &THEN KILL aspectSE all
&IF [EXIST aspectSW -GRID] &THEN KILL aspectSW all
&IF [EXIST aspectNW -GRID] &THEN KILL aspectNW all
```

```
&IF [EXIST aspectN -GRID] &THEN KILL aspectN all
&IF [EXIST aspectE -GRID] &THEN KILL aspectE all
&IF [EXIST aspectS -GRID] &THEN KILL aspectS all
&IF [EXIST aspectW -GRID] &THEN KILL aspectW all
```

```
&IF [EXIST aspectFlat -GRID] &THEN KILL aspectFlat all
```

```
/*&IF [EXIST AspectCov -COVER] &THEN KILL AspectCov ALL
/*GRIDPOLY aspectClass AspectCov
&RETURN
```

```
/*****
```

```
&ROUTINE Slope
```

```
&IF [EXIST SlopeClass -GRID] &THEN KILL SlopeClass ALL
```

```
&r d:\renee_gis\scripts\checkproggrid.txt
```

```
&TYPE set window to dem25int...
SETWINDOW dem25int
SETMASK dem25int
&TYPE setmask complete
```

```
&TYPE where the condition is false the value assigned will be 0
&TYPE Where the condition is true the value assigned will be the upper limit of the class range
```

```
slope04 = con(slopeint ge 0 and slopeint le 4, 4, 0)
  &TYPE slope range 0 - 4 done
slope08 = con(slopeint gt 4 and slopeint le 8, 8, 0)
  &TYPE slope range 5 - 8 done
slope14 = con(slopeint gt 8 and slopeint le 14, 14, 0)
  &TYPE slope range 9 - 14 done
slope20 = con(slopeint gt 14 and slopeint le 20, 20, 0)
  &TYPE slope range 15 - 20 done
slope30 = con(slopeint gt 20 and slopeint le 30, 30, 0)
  &TYPE slope range 21 - 30 done
slope45 = con(slopeint gt 30 and slopeint le 45, 45, 0)
  &TYPE slope range 31 - 45 done
slope90 = con(slopeint gt 45, 90, 0)
  &TYPE slope > 45 done
```

```
SlopeClass = (slope04 + slope08 + slope14 + slope20 + slope30 + slope45 + slope90)
```

```
&TYPE Slope Done
SETMASK off
Q
```

```

&TYPE Delete Temporary GRIDs
&IF [EXIST slope04 -GRID] &THEN KILL slope04 all
&IF [EXIST slope08 -GRID] &THEN KILL slope08 all
&IF [EXIST slope14 -GRID] &THEN KILL slope14 all
&IF [EXIST slope20 -GRID] &THEN KILL slope20 all
&IF [EXIST slope30 -GRID] &THEN KILL slope30 all
&IF [EXIST slope45 -GRID] &THEN KILL slope45 all
&IF [EXIST slope90 -GRID] &THEN KILL slope90 all

/*&IF [EXIST SlopeCov -COVER] &THEN KILL SlopeCov ALL
/*GRIDPOLY SlopeClass SlopeCov
&RETURN
/*****

&ROUTINE Elevation

&IF [EXIST ElvtnClass -GRID] &THEN KILL ElvtnClass ALL

&r d:\renee_gis\scripts\checkproggrid.txt

&TYPE set window to dem25int...
SETWINDOW dem25int

SETMASK dem25int
&TYPE setmask complete

&TYPE where the condition is false the value assigned will be 0
&TYPE Where the condition is true the value assigned will be the upper limit of the class range

Elvtn25 = con(dem25int ge 0 and dem25int le 25, 25, 0)
  &TYPE Elevation 0 - 25 range done
Elvtn50 = con(dem25int gt 25 and dem25int le 50, 50, 0)
  &TYPE Elevation 26 - 50 range done
Elvtn100 = con(dem25int gt 50 and dem25int le 100, 100, 0)
  &TYPE Elevation 51 - 100 range done
Elvtn200 = con(dem25int gt 100 and dem25int le 200, 200, 0)
  &TYPE Elevation 101 - 200 range done
Elvtn300 = con(dem25int gt 200 and dem25int le 300, 300, 0)
  &TYPE Elevation 201 - 300 range done
Elvtn400 = con(dem25int gt 300 and dem25int le 400, 400, 0)
  &TYPE Elevation 301 - 400 range done
Elvtn550 = con(dem25int gt 400 and dem25int le 550, 550, 0)
  &TYPE Elevation 401 - 550 range done
Elvtn2800 = con(dem25int gt 550, 2800, 0)
  &TYPE Elevation > 550 range done

ElvtnClass = (Elvtn25 + Elvtn50 + Elvtn100 + Elvtn200 + Elvtn300 + Elvtn400 + Elvtn550 + Elvtn2800)

&TYPE Elevation Done
SETMASK off

```

Q

```
&TYPE Delete Temporary Elevation GRIDs
&IF [EXIST Elvtn25 -GRID] &THEN KILL Elvtn25 all
&IF [EXIST Elvtn50 -GRID] &THEN KILL Elvtn50 all
&IF [EXIST Elvtn100 -GRID] &THEN KILL Elvtn100 all
&IF [EXIST Elvtn200 -GRID] &THEN KILL Elvtn200 all
&IF [EXIST Elvtn300 -GRID] &THEN KILL Elvtn300 all
&IF [EXIST Elvtn400 -GRID] &THEN KILL Elvtn400 all
&IF [EXIST Elvtn550 -GRID] &THEN KILL Elvtn550 all
&IF [EXIST Elvtn2800 -GRID] &THEN KILL Elvtn2800 all
```

```
/*&IF [EXIST ElvtnCov -Cover] &THEN KILL ElvtnCov
/*GRIDPOLY ElvtnClass ElvtnCov
&RETURN
/*****
```

```
&ROUTINE DEM_Bnd
```

```
&IF [EXIST DEMBounds -GRID] &THEN KILL DEMBounds all
&r d:\renee_gis\scripts\checkproggrid.txt
&TYPE set window to dem25int...
SETWINDOW dem25int
SETMASK dem25int
&TYPE setmask complete
```

```
&TYPE where the condition is false the value assigned will be 0
&TYPE Where the condition is true the value assigned will be 1
```

```
DEMBounds = con(dem25int le 2800, 1, 0)
&TYPE DEMBounds DEM boundary created
SETMASK off
```

Q

```
&IF [EXIST DEM_Bnd -Cover] &THEN KILL DEM_Bnd
GRIDPOLY DEMBounds DEM_Bnd
&RETURN
/*****
```

```
&ROUTINE DEMBnd_Shp
ARCSHAPE DEM_Bnd POLYS DEM_Bndry
&RETURN
/*****
```

```
&ROUTINE BndandTaupo
/* Region Boundary with Lake Taupo clipped out (final Region grid)
&IF [EXIST MyBnds -GRID] &THEN
  &TYPE The GRID MyBnds exists
&ELSE &DO
  &TYPE the coverage MyBnds does not already exist. Creating now...
  &R D:\renee\scripts\checkproggrid.txt
  MyBnds = shapegrid(d:\renee_gis\input_data\bndsandLake, Grid_Code, 25)
```

Q
&TYPE MyBnds GRID created
&END
&RETURN